
9.2 CONTOUR (cycle 14, DIN/ISO : G37)

Attention lors de la programmation!
Dans le cycle 14 CONTOUR, listez tous les sous-programmes qui
doivent être superposés pour former un contour entier.

Ce cycle ne peut être exécuté que dans les modes
d'usinage FUNCTION MODE MILL et FUNCTION MODE
TURN .
Le cycle 14 est actif avec DEF, ce qui signifie qu'il est
actif dès qu'il est défini dans le programme CN.
Vous pouvez lister jusqu'à 12 sous-programmes
(contours partiels) dans le cycle 14.

Paramètres du cycle
Numéros de labels du contour : entrer tous
les numéros de labels des différents sous-
programmes qui doivent être superposés à
un contour. Confirmer chaque numéro avec
la touche ENT. Mettre fin aux saisies avec la
touche END Saisie des numéros de 12 sous-
programmes max., de 1 à 65 535

9.3 Contours superposés

Principes de base
Un nouveau contour peut être construit en superposant des poches
et des îlots. De cette manière, vous pouvez agrandir la surface d'une
poche par superposition d'une autre poche ou la réduire avec un îlot.

Exemple

12 CYCL DEF 14.0  CONTOUR

13 CYCL DEF 14.1  LABEL
CONTOUR1/2/3/4

Sous-programmes : poches superposées

Les exemples suivants sont des sous-programmes de
contours qui sont appelés dans un programme principal
du cycle 14 CONTOUR.

Les poches A et B se superposent.
La commande calcule les points d'intersection S1 et S2. Ils n'ont
pas besoin d'être programmées.
Les poches sont programmées comme des cercles entiers.

Sous-programme 1: Poche A

51 LBL 1

52 L  X+10 Y+50  RR

53 CC  X+35 Y+50

54 C  X+10 Y+50  DR-

55 LBL 0

Sous-programme 2: Poche B

56 LBL 2

57 L  X+90 Y+50  RR

58 CC  X+65 Y+50

59 C  X+90 Y+50  DR-

60 LBL 0

Surface „d'addition“
Les deux surfaces partielles A et B, y compris leurs surfaces
communes, doivent être usinées :

Les surfaces A et B doivent être des poches.
La première poche (dans le cycle 14) doit débuter à l’extérieur
de la seconde

Surface A :

51 LBL 1

52 L  X+10 Y+50  RR

53 CC  X+35 Y+50

54 C  X+10 Y+50  DR-

55 LBL 0

Surface B :

56 LBL 2

57 L  X+90 Y+50  RR

58 CC  X+65 Y+50

59 C  X+90 Y+50  DR-

60 LBL 0

Surface „de soustraction“
La surface A doit être usinée sans la partie recouverte par B:

La surface A doit être une poche et la surface B, un îlot.
A doit débuter à l’extérieur de B.
B doit commencer à l'intérieur de A

Surface A :

51 LBL 1

52 L  X+10 Y+50  RR

53 CC  X+35 Y+50

54 C  X+10 Y+50  DR-

55 LBL 0

Surface B :

56 LBL 2

57 L  X+40 Y+50  RL

58 CC  X+65 Y+50

59 C  X+40 Y+50  DR-

60 LBL 0

Surface „d'intersection“
La surface commune de recouvrement de A et de B doit être
usinée. (Les surfaces sans recouvrement ne doivent pas être
usinées.)

A et B doivent être des poches.
A doit commencer à l’intérieur de B.

Surface A :

51 LBL 1

52 L  X+60 Y+50  RR

53 CC  X+35 Y+50

54 C  X+60 Y+50  DR-

55 LBL 0

Surface B :

56 LBL 2

57 L  X+90 Y+50  RR

58 CC  X+65 Y+50

59 C  X+90 Y+50  DR-

60 LBL 0

